
ABSTRACT  

Effect size estimates are used to measure, in a standardized manner, the magnitude of treatment effect or 
association between two or more variables. The standardized mean difference (Cohen’s  -statistic) between two or 

more conditions is a frequently used measure of effect size in both the social and biomedical sciences. In this 
situation, the response (dependent) variable is continuous, whereas the predictor (independent) variable is 
categorical. Although some web-based and Microsoft EXCEL-based software are available to calculate effect size 
using summary statistics, available software to calculate both effect size and  corresponding confidence interval are 
limited. Moreover, to the authors’ knowledge, there is no standard SAS® procedure that directly produces the  -

statistic and  confidence interval. Therefore, in this paper, we describe calculation of the  -statistic and its confidence 

interval directly from raw survey data following the methods of Morris and DeShon (2002), Nakagawa and Cuthill 
(2007), and Fritz, Morris and Richler (2012). The SAS Output Delivery System (ODS) is used to produce  the  -

statistic and its confidence interval. Three types of survey designs are considered: (a) Independent group single post-
test (completely randomized) design; (b) Within-subject (one-group pretest–posttest) design, and (c) Independent-
groups pretest–posttest design (IGPP). Finally, the authors present a SAS macro that implements the methodology 
for the   statistic and confidence interval for the three designs mentioned above. A basic understanding of SAS is 

sufficient to be able to use this macro, and without skill required in SAS macro programming. When the macro is 
invoked, a table with summary statistics, effect size, and confidence interval are calculated, and the results can be 
saved in Microsoft EXCEL format. This macro has been tested on Windows SAS 9.2. 

Keywords: effect size, Cohen’s  -statistic, confidence interval,  Pre-Post Design, SAS macro 

INTRODUCTION  

Effect size estimates are used to measure, in a standardized manner, the magnitude of treatment effect or 
association between two or more variables. Cohen’s   statistic (Cohen, 1988) is a widely used measure of effect size 

to compare two or more conditions, treatment versus control, for example. Within research settings in the biomedical 
and social sciences, three common designs to evaluate treatment effects or associations include: (a) Independent 
group single post-test (completely randomized) design; (b) Within-subject (one-group pretest–posttest) design, and 
(c) Independent-groups pretest–posttest design (IGPP). Whereas some professional journals explicitly require 
authors to supply effect sizes and corresponding confidence intervals (e.g. in addition to p-values) for results of 
comparisons of two or more groups (e.g. treatment effects), most published research articles do not provide 
standardized effect sizes. Moreover, the reporting of confidence intervals for estimates of effect size is even less 
frequent. Of note, providing estimates of effect size, whether for treatment comparisons of associations between 
variables, provides the direction and strength of association in a standard metric that is easily interpretable and also 
useful for meta-analysis. Perhaps a main reason for limited reporting of effect sizes is simply the lack of availability 
within popular statistical software packages. In general, researchers typically rely on a manual approach to calculate 
effect sizes and corresponding confidence intervals. Indeed, to our knowledge even in SAS, there is no standard 
procedure (PROC) that produces effect size (Cohen’s d) along with its confidence interval. 

Therefore, the primary purpose of this paper is to present a SAS macro that is used to calculate effect size and 
confidence interval from raw survey data (not summary statistics) collected using various study designs including: (a) 
Independent group single post-test design (IG design); (b) Within-subject (one-group pretest–posttest) design (RM 
design); and (c)  Independent-group pre-test post-test design (IGPP design). In the following sections we discuss the 
data structure requirements and methods (formulas) used to calculate the effect size and its confidence interval for 
each design. Finally, we illustrate an example on how to invoke (call) the SAS macro to calculate the effect size and 
corresponding confidence interval. 

EXAMPLE DATA: SURVEY DATA ON PSYCHOLOGICAL STATUS 

To best explain the calculation of effect size (   and its confidence interval, we use the following long format SAS 

data set from a fictitious study of psychological status. The snapshot of the psychological survey data is shown in 
Table 1 and includes only data from the first five subjects. In this data set, variable ID is used to identify the subject; 
VISIT variable has two values consisting of 1 (pre) and 2 (post); and the GROUP variable has two values consisting 
of 1(treatment) and 2(control). In addition, there are three continuous (dependent) variables for which we want to 
calculate the effect size. These three dependent variables are fictitious total scores on the 17-item PTSD-Checklist 

(PCL), total score on the 18-item Brief Symptom Inventory (BSI), and total score on the 32-item self-reported Trauma 
Related Guilt Inventory (TRGI).  

Within this framework, the following are required for operation of the SAS macro:  (a) data must be in long format, as 

shown in the  table 1; (b) all variables must be in numeric format; (c) the GROUP variable must have values 1 and 2 
(e.g. treatment and control); and (d) the TIME variable (VISIT in the following example) must be specified as only 2 
values (e.g. 1 and 2, or 1 and 3), although the data set can have more than two values (e.g. visit 1,2,3, etc). To 
calculate the effect size for the independent group single post-test design (IG design), we ignore the VISIT variable 

from the data set. Similarly, for the within-subject (one-group pretest–posttest) design (RM design),  we ignore the 
GROUP variable. For the independent group pre-test posttest design (IGPP design), we use all columns from the 
data set. An efficient  feature of this macro is that the effect size of more than one continuous dependent variable (for 
example PCL, BSI, and TRGI scores) can be simultaneously calculated.  

 
 

ID GROUP VISIT PCL BSI TRGI 

2001 1 1 80 40 15 

2001 1 2 74 41 14 

2002 1 1 79 58 15 

2002 1 2 39 13 6 

2003 1 1 67 33 3 

2003 1 2 59 38 3 

2004 2 1 61 27 10 

2004 2 2 55 26 13 

2005 2 1 45 9 0 

2005 2 2 37 9 0 

Table 1: Example of Raw Survey Data  

DESIGN 1: INDEPENDENT- GROUPS POST-TEST DESIGN (IG) 

In the case of the independent-groups (IG) single post-test design, the effect size (Cohen’s  ) is simply the difference 

between the mean of the treatment group (      and control group (      divided by the standard deviation. In 

calculating the effect size, the macro assumes homogeneity of variances (i.e. both the control and treatment groups 
have the same underlying variance). The natural choice of variance in this case is the pooled standard deviation (   , 

which can be calculated using the equation (1) below. 

This pooled standard deviation,   , is used to calculate the effect size using  equation (2) below.   

where    and     are sample means of group 1 (control group) scores and group 2 (treatment group) scores, 

respectively. Similarly,   
  and   

  are sample variances and       and    are sample sizes for each group. Invariably, it 
is also important to report a confidence interval for the effect size in addition to the point estimate.  A confidence 
interval for this type of independent group single measure design can be calculated using equation (3) below. 

 
 

     
 (      

 

            

 

    ) 
(3) 

where     is the estimate of standard error of the effect size (Hunter and Schmidt 2004; Nakagawa and Cuthill (2007) 
and is calculated using equation (4) below. 

where      is the appropriate two-tailed value for the standard normal distribution (for example            for     

confidence interval) and      
 is the standard error of the  -statistic computed using equation (2). 

Even though equation (1) calculates the   value by subtracting the mean of group 1 from the mean of group 2, the 
sign of d is arbitrary. When the direction of the result is consistent with the a priori hypothesis, it should be reported 
with a positive sign. When a result opposite the direction of that specified by the hypothesis, it should reported with a 
negative sign. This is easily accomplished by designating the group that is expected to have the higher mean as 
group 2. 

To calculate the effect size and its confidence interval for this design, the macro is invoked (called), as in the following 
example. The user can replace all the parameter values, as in their data set. For example, SAS data 
SESUG12.survey_data can be replaced by their own SAS data. Details for calling the SAS macro are provided in the 
next section. 

 

DESIGN 2: ONE GROUP PRETEST–POSTTEST DESIGN (RM) 

Paired sample t–tests compare group means when the two groups are related in various research designs, such as 
matched pairs and repeated measures. If a single group dependent variable is measured before and after treatment, 
the effect size (Cohen’s  ) is simply the mean difference between the pre- and post-test scores divided by the sample 

standard deviation of the change score, as provided in equation (5) per Morris and DeShon (2002)  

     
     

  
  (5) 

where    represents post treatment mean,    represents pre-treatment mean, and    represents the sample standard 

deviation of the change scores. Becker (1988) provided the following equation (6) to calculate the standard error of 
the effect size (Cohen’s   statistic) for the paired design, which can be used to calculate a confidence interval. 

The confidence interval for effect size   can be calculated using confidence interval equation (3) from the previous 
section replacing the corresponding values of effect size (   and standard error of effect size for the paired design. In 

this equation,     is the correlation coefficient between the pre-test and post test scores. 

Cohen (1988) suggested that calculation of effect size for the dependent paired group can be achieved using the 
same equation as the independent group design described above. In this case, the pooled standard deviation is 
calculated by equation (7) below.   

    √
  
    

 

 
 (7) 

With this macro, users have the option to calculate effect size and its confidence interval using one of the two 
standard deviations (standard deviations of raw scores, or standard deviation of change score) in the denominator of 
equation (5). To calculate the effect size using the raw standard deviation, the user will use design 1 (design=1); to 
calculate the effect size using the standard deviation of the difference scores, the user will use design=2 when calling 
the SAS macro. This can be accomplished by use of the syntax example below. . 

 

Of note, the effect size calculation that uses the change score standard deviation is always greater than the effect 
size calculation from the raw standard deviation which suggests that the within-subject design is more powerful than 
the between-subject design (Feingold, 2009). Feingold (2009) also suggests that the effect size calculated using the 
raw standard deviation conveys the treatment potency. 

DESIGN 3: INDEPENDENT-GROUPS PRETEST–POSTTEST DESIGN (IGPP)  

In the case of the independent-group pretest-posttest design, different groups receive different treatments (e.g., 
experimental and control groups), and outcomes are measured both before and after the treatment. In this case, it is 
natural to calculate the effect size within each treatment condition, and then subtract the control group effect size 
from the experimental-group effect size as in equation (8) to compute the overall effect size (Becker, 1988). 

The effect size for each treatment condition is defined as the pretest–posttest change in mean score divided by the 
standard deviation of raw score or change scores, as in the one group pre-post (RM) discussed in the previous 
section.  

       
              

  
  - 

              

  
 (8) 

where     and     are the standard deviation of treatment group scores and control group scores, respectively. Both 
standard deviations (   and   ) can be calculated either using raw scores (equation 7) or the change score of each 

group (treatment and control), as in the previous section. 

Because of the complexity in calculating sampling variances of effect size estimates for the independent group pre-
post (IGPP) design, only the approximation to the variance of the effect size is calculated, specifically, the sum of the 
variances for the two components (Becker, 1988; Morris and DeShon, 2002).  The variance of each component of 
effect size can be calculated using equation (4) for the raw score case, and equation (6) if the change score standard 
deviation is used. This can be conducted for the treatment and control groups separately. The standard error of effect 

size for the IGPP design is simply the square root of the sum of variances (i.e.         
     (      ). The 

confidence interval for effect size       can be calculated using equation (3) in the previous section replacing effect 

size and standard error from the      design. 

In this IGPP design, as in the case of the within-subject repeated measure (RM) design, users have two options to 
calculate effect size and confidence interval using either the raw standard deviation or the standard deviation of the 
change scores. If the user wants to calculate the effect size and confidence interval using the raw standard deviation, 
design=1 should be selected as follows.  

 

If design = 1 is replaced by design=2 (keeping everything else the same) in the above macro call, the standard 
deviation of the change score will be used to calculate the effect size and confidence interval. As in the within-subject 
repeated measures design in the previous section, the effect size will always be higher if the standard deviation of the 
difference scores (post minus pre scores) is used instead of the raw scores (i.e. pooled standard deviation of pre-
score and post-scores). 

EFFECT SIZE CALCULATION MACRO PARAMETERS 

The following are the macro parameters used for the effect size macro. 

data = name of SAS data set (long format) 
idvar = name of the variable for subject id  
groupvar = name of group variable name (treatment/control group). If the design is single  group (within-subject) 

pre-post or repeated measure (RM) design, leave this blank (values must be 1 or 2). 
timevar = name of time sequence variable (e.g. Visit=1 is pre and visit=2 is post). If the design is a single 

measure two group independent design (IG), leave this blank 
respvars = name of the continuous dependent variable(s) for which to calculate the effect size(s). If there are 

more than one dependent variables, use a single space between each variable.  
timeval1 = Values of time variable in the pre-visit (1 for pre-test for example) 
timeval2 = Values of time variable in the post-visit (2 for post-visit for example) 
cl = confidence level (e.g. 0.95 to calculate 95% confidence interval) 

HOW TO INVOKE THE SAS MACRO 

Calling the effect size macro (provided in the Appendix) is straight-forward. The user specifies the library reference, 
dataset, and other parameters (idvar, groupvar, etc.) described above. Some parameters are specific to the three 
study designs (IG, RM, IGPP). For example, in the one group pretest-posttest design, macro parameter  groupvar 
can be kept blank (i.e. groupvar = ) or eliminated from the macro call (i.e. %effect_size statement). Once the macro 
codes are submitted (compiled), the macro can be invoked using the statement %effect_size, as in the example 

below. A table with summary statistics (sample size, mean, and standard deviation), effect size, and its confidence 
interval are calculated, and the results can be saved in Microsoft EXCEL.  
 

Consider the case where we want to compare the pre- and post-treatment scores of several dependent variables 
(BDI, BSI, and TRGI scores) for a single group within-subject pre-post design. By supplying the  macro call below, the 
effect size, 95% confidence interval,  and summary statistics for the pre- treatment and post-treatment scores will be 
calculated (Table 2). Depending upon the study design, this table format will vary some but the effect size and 
confidence interval will always be produced.  In the example below, the parameter data has the value 
SESUG12.survey_data. Please note in the following macro call (%effect_size) that SESUG12 is the SAS library 
name, and survey_data is SAS data set name. In this case, the macro parameter groupvar is ignored from the 

parameter list, as we are interested in calculating the effect size for the single group within-subject pre-post or 
repeated measure (RM) design. The first column of the results table (column name measure) indicates the direction 
of the effect size calculated. In this example, PCL2-PCL1 (PCL2 minus PCL) indicates that the pre-test mean is 
subtracted from the post-test mean to calculate the numerator of the effect size. 

 
 

Response Sample 
Size (n) 

Pre-test 
Mean 

Pre-test 
SD 

Post-test 
Mean 

Post-test 
SD 

Effect 
Size (d) 

95% Conf. 
Interval 

PCL2-PCL1 29 58.6 16.41 48.62 19.41 -0.556 (-0.870,-0.242) 

BSI2-BSI1 29 28.07 17.48 17.41 17.41 -0.611 (-1.002,-0.220) 

TRGI2-TRGI1 30 7.73 4.95 6.63 4.75 -0.227 (-0.546,0.093) 

Table 2: Effect Size For Single Group Pre-Test/Post Test Design 

CONCLUSION  

The calculation of effect size and corresponding confidence interval based on a user-friendly SAS macro is 
presented. This macro can be used to calculate the effect size for one or more continuous response (dependent) 
variables for three types of study designs; (a) Independent group single post-test (completely randomized) design; (b) 
Within-subject (single-group pretest–posttest) design; and (c) Independent-group pretest–posttest design (IGPP). 
Individuals with a basic working knowledge of SAS programming and statistical understanding of effect size 
calculation (and interpretation) should be able to use this macro with minimal difficulty. A requirement of this macro is 
that data set has to be structured in the long format, and values of the group variable and time variable have to be 

coded with values of 1 and 2 respectively. In addition, a limitation of this macro is that  small sample bias correction is 
not implemented when calculating the effect size and confidence interval.  

 APPENDIX B: SAS MACRO 

****************************************************************************; 

** Brief Introduction to Macro:             ** 

** --------------------------             ** 

** Purpose of this macro is to calculate effect size (Cohen's d)    ** 

** and its confidence interval as well as summary statistics     ** 

** (sample size, mean, standard deviation) for one or more continuous  ** 

** dependent variables . Three types of designs are considered:     ** 

**  (a) Independent group post-test (completely randomized) design (IG) ** 

**  (b) Within-subject (one-group pretest–posttest) design, and (RM)  ** 

**  (c) Independent-groups pretest–posttest design (IGPP)      ** 

**  In case of designs (b) and (c), two types of standard deviations   ** 

** (denominator of effect size formula) are used to calculate effect size  ** 

**  (i) using raw standard deviation or (ii) using standard deviation  ** 

** of difference scores( difference of pre post scores).      ** 

**                     ** 

**  How to Cite this Macro:              ** 

** ----------------------              ** 

** Kadel R. P., Kip K. E. (2012). A SAS Macro to Compute Effect Size   ** 

** (Cohen’s  ) and its Confidence Interval from Raw Survey Data.    **  

** Proceedings of the annual Southeastern SAS Users Group Conference.    ** 

** --------------------------------------------------------------------------------** 

** Contact Address for comments/Feedback/Suggestions about Macro:     ** 

** -------------------------------------------------------------    ** 

**  Rajendra Kadel, at rkadel@health.usf.edu        ** 

**                     ** 

****************************************************************************;  

%macro  effect_size( data,idvar,design,groupvar, timevar, 

     timeval1, timeval2,respvars,cl=0.95); 

 ************************************************: 

 ** case I: independent group post test design***; 

 ** In this case timevar will be blank***********; 

 ************************************************; 

 data _null_; 

  call symput("conf","&cl"*100); 

 run; 

 %if &timevar= %then %do; 

  proc sql noprint; 

   select distinct(&groupvar) into :grp1-:grp2 from &data; 

  quit; 

  proc sort data=&data out=data1(keep=&idvar &groupvar &respvars); 

   by &groupvar; 

  run; 

  **** mean ****; 

  ods trace on; 

  proc means data=data1 mean; 

  by  &groupvar; 

  var  &respvars; 

  ods output Summary=Summary_mean; 

  run; 

  ods trace off; 

  **** sd ****; 

  proc means data=data1 std; 

  by  &groupvar; 

  var &respvars ; 

  ods output Summary=Summary_std; 

  run; 

  **** N ****; 

  proc means data=data1 n; 

  by  &groupvar; 

  var &respvars; 

mailto:rkadel@health.usf.edu

  ods output Summary=Summary_n; 

  run; 

  ** transpose data***; 

  proc transpose data=summary_mean out=mean_long(rename=(_LABEL_=static))  

   name=Measure prefix=&groupvar; 

   id &groupvar; 

  run; 

  proc transpose data=summary_std out=std_long(rename=(_LABEL_=static))  

   name=Measure prefix=&groupvar; 

   id &groupvar; 

  run; 

  proc transpose data=Summary_n out=n_long(rename=(_LABEL_=static))  

   name=Measure prefix=&groupvar; 

   id &groupvar; 

  run; 

  ** merge data***; 

  data combined_withingroup_IG; 

  merge mean_long(keep=measure &&groupvar.&grp1 &&groupvar.&grp2  

      rename=(&&groupvar.&grp1=mean&grp1 &&groupvar.&grp2=mean&grp2))  

   std_long(keep=&&groupvar.&grp1 &&groupvar.&grp2  

      rename=(&&groupvar.&grp1=sd&grp1 &&groupvar.&grp2=sd&grp2))  

   n_long(keep=&&groupvar.&grp1 &&groupvar.&grp2  

      rename=(&&groupvar.&grp1=n&grp1 &&groupvar.&grp2=n&grp2)) ; 

  run; 

  *** Calculate effect size and CI***; 

  *** S.E. of effect size calculate using eqn 16 of Nakagawa & Cuthill (2007)*; 

  data effectsize_table_IG; 

   retain measure n&grp1 mean&grp1 sd&grp1 n&grp2 mean&grp2  

    sd&grp2 eff_size ci_low ci_up; 

   set combined_withingroup_IG; 

   M_Change=(mean&grp2-mean&grp1); 

   SD_pool=sqrt((((n&grp1-1)*sd&grp1**2)+ 

      ((n&grp2-1)*sd&grp2**2))/(n&grp1+n&grp2-2)); 

   eff_size=M_Change/SD_pool; 

   ** Standard error of d**; 

   se_d1=((n&grp1+n&grp2-1)/(n&grp1+n&grp2-3)); 

   se_d2= 4/(n&grp1+n&grp2); 

   se_d3=1+eff_size**2/8; 

   se_d= sqrt(se_d1*se_d2*se_d3); 

   z=PROBIT(.50+&cl/2);  

   ci_low=eff_size-z*se_d; 

   ci_up=eff_size+z*se_d; 

      format _NUMERIC_ 8.3; 

   format n&grp1 n&grp2 8.0 mean&grp1 sd&grp1 mean&grp2 sd&grp2 8.2; 

  drop se_d1 se_d2 se_d3 z se_d M_Change SD_pool; 

  run; 

 %goto printresult; 

 %end; 

 

 ****************************************************************: 

 ** case II: One group pre-post or repeated measure (RM) design**; 

 ** In this case Group variable will be blank********************; 

 ****************************************************************; 

 %macro eff_size_paired(data=,idvar=,design=2,timevar=,timeval1=, 

       timeval2=,respvars=,cl=,outdata=); 

  ** create macro variables***; 

  data _null_ ; 

   call symput('nresp',countw("&respvars")); 

  run; 

  %let respvars=&respvars; 

  %let nresp=&nresp; 

  data _null_; 

   %do i=1 %to %eval(&nresp); 

    call symputx("resp&i",scan("&respvars",&i,'')); 

   %end; 

  run; 

  *** create table with distinct values***; 

  proc sort data=&data out=data2(keep=&idvar &timevar &respvars) nodupkey; 

  by &idvar &timevar; 

  run; 

  ** make data wide***; 

  %do i=1 %to %eval(&nresp); 

   Proc SQL; 

    create table survey_wide&i as  

    select distinct a.&idvar, 

    (Select b.&&resp&i from data2 as b where b.&timevar=&timeval1  

     and a.&idvar=b.&idvar) as &&resp&i..&timeval1, 

    (Select c.&&resp&i from data2 as c where c.&timevar=&timeval2  

     and a.&idvar=c.&idvar ) as &&resp&i..&timeval2, 

    calculated &&resp&i..&timeval2 - calculated &&resp&i..&timeval1  

     as &&resp&i.._diff from data2 as a where a.&timevar=&timeval2; 

   quit; 

 

   ods trace on; 

   proc corr data=survey_wide&i pearson; 

   var &&resp&i..&timeval1 &&resp&i..&timeval2; 

   ods output  Corr.PearsonCorr=corr; 

   run; 

   ods trace off; 

   * create macro variable for correlation**; 

   proc sql noprint; 

    select &&resp&i..&timeval2 into :r   

    from work.corr where variable="&&resp&i..&timeval1"; 

   quit; 

   * create table for effect size (using SD of difference score)******; 

   ** SE of effect size calculate using Nakagawa & Cuthill,2007 eq(18))*; 

   %if &design=2 or &design=  %then %do; 

    proc sql ; 

     create table eff_table&i as  

     select min(count(&&resp&i..1), 

     count(&&resp&i..&timeval2)) as  n , 

     mean(&&resp&i..1) as mean1, 

     std(&&resp&i..&timeval1) as sd1 format=8.2, 

     mean(&&resp&i..2) as mean2, 

     std(&&resp&i..&timeval2) as sd2 format=8.2, 

     mean(&&resp&i..&timeval2-&&resp&i..&timeval1)  

      as mean_dif format=8.2,  

     std(&&resp&i.._diff) as sd_dif format=8.2, 

     (calculated mean_dif) /(calculated sd_dif) format=8.3  

      as effect_size from survey_wide&i; 

    quit; 

   %end; 

   ** cohen (1988) suggested to use pooled standard deviation from raw scores 

; 

   %if &design=1 %then %do; 

    proc sql ; 

     create table eff_table&i as  

     select min(count(&&resp&i..1), 

     count(&&resp&i..&timeval2)) as  n , 

     mean(&&resp&i..1) as mean1,std(&&resp&i..&timeval1) as sd1, 

     mean(&&resp&i..2) as mean2,std(&&resp&i..&timeval2) as sd2, 

     mean(&&resp&i..&timeval2-&&resp&i..&timeval1) as mean_dif, 

      std(&&resp&i.._diff) as sd_dif format=8.2, 

sqrt(((calculated sd1)**2*(count(&&resp&i..1)-1)+(calculated 

sd2)**2* 

(count(&&resp&i..2)-1))/(count(&&resp&i..1)+count(&&resp&i..2)-2))  

as sd_pool,  

(calculated mean2- calculated mean1) /(calculated sd_pool) 

format=8.3 as effect_size 

     from survey_wide&i; 

    quit; 

   %end; 

   *create table for each response**; 

   data eff_table&i; 



SESUG2012 

10 

retain Response n mean1 sd1 mean2 sd2 tval pval effect_size ci_low 

ci_up; 

    length Response $10.; 

    set eff_table&i; 

    Response="&&resp&i"; 

    se_d= sqrt(2*(1-&r)/n+(effect_size**2)/(2*(n-1)));  

    /* z=quantile('NORMAL',.975); also gives same*/ 

    z=PROBIT(.50+&cl/2); /* probit returns quantiles (zscore) */ 

    ci_low=effect_size-z*se_d; 

    ci_up=effect_size+z*se_d; 

       format _NUMERIC_ 8.3; 

    format n 8.0 mean1 sd1 mean2 sd2 tval 8.2; 

    *drop se_d t mean_dif sd_dif; 

    drop z; 

   run; 

 

  %end; 

  *** collecting effect size and confidence interval for all values***; 

  data &outdata; 

   set  

   %do i=1 %to %eval(&nresp); 

   eff_table&i 

   %end;; 

  run; 

 %mend eff_size_paired; 

 *** to calculate effect size for paired t-test design***; 

 %if &groupvar= %then %do; 

  %eff_size_paired(data=&data,idvar=&idvar,design=&design, 

   timevar=&timevar,timeval1=&timeval1, timeval2=&timeval2, 

   respvars=&respvars,cl=&cl,outdata=effectsize_table_RM) 

 %put "Design &design"; 

 %end; 

 *****************************************************************; 

 ** case II: Independent Group Pre-Post (IGPP)design**************; 

 ** Provide *rovide the name of group and time variable***********; 

 *****************************************************************; 

 %if &groupvar^= and &timevar^= %then %do; 

   proc sql noprint; 

   select distinct(&groupvar) into :grp1-:grp2 from &data; 

  quit; 

  ** to calculate effect size of 2-group pre-post design***; 

  ** make two separate data set for group 1 and group 2****; 

  data data_&grp1(keep=&idvar &timevar &respvars) data_&grp2(keep=&idvar  

     &timevar &respvars); 

   set &data; 

   if &groupvar=&grp1 then output data_&grp1; 

   if &groupvar=&grp2 then output data_&grp2; 

  run; 

  * Noow calculate effect size for both groups*********************; 

  * using either design 1(independent) or design 2 paired desig)***; 

  %eff_size_paired(data=data_&grp1,idvar=&idvar,design=&design, 

   timevar=&timevar,timeval1=&timeval1, timeval2=&timeval2, 

   respvars=&respvars,cl=&cl,outdata=effectsize_table&grp1) 

  %eff_size_paired(data=data_&grp2,design=&design,idvar=&idvar, 

   timevar=&timevar,timeval1=&timeval1, timeval2=&timeval2, 

   respvars=&respvars,cl=&cl,outdata=effectsize_table&grp2) 

 

  * combine two tables to get effect size***; 

  

  * IGPP design is square root of sum of variance ****; 

  proc sql; 

  create table effectsize_table_IGPP as  

 select  trim(a.Response) ||"&grp1"||"-"||trim(a.Response)||"&grp2" as 

Response,   

   a.n as n_&grp1, 

   a.mean1 as M1_&grp1 format 8.2, a.sd1 as SD1_&grp1 format=8.2, 

   a.mean2 as M2_&grp1 format=8.2, a.sd2 as SD2_&grp1 format=8.2, 

charbel
Highlight

   b.n as n_&grp2,   

   b.mean1 as M1_&grp2 format=8.2, 

   b.sd1 as SD1_&grp2 format=8.2, b.mean2 as M2_&grp2 format=8.2, 

   b.sd2 as SD2_&grp2 format=8.2, a.effect_size - b.effect_size as eff_size, 

 calculated eff_size-PROBIT(.50+&cl/2)*(sqrt(a.se_d**2+b.se_d**2)) as 

CI_LOW, 

calculated eff_size+PROBIT(.50+&cl/2)* (sqrt(a.se_d**2+b.se_d**2)) as 

CI_UP 

  from work.effectsize_table&grp1 as a, work.effectsize_table&grp2 as b 

  where a.Response=b.response; 

  quit; 

 %goto printresult; 

 %end; 

 ** calculate macro variable for confidence level***; 

 

 %printresult: 

 ** Print results***; 

 ods html file="survey_data_eff_size.xls"; 

 %if &timevar= %then %do; 

  title "Table 1: Effect Size (Cohen'd) and its Confidence Interval  

    for Independnt Group Post-Test Design(IG)."; 

  proc sql nonumber; 

  select  trim(measure) ||"&grp2"||"-"||trim(measure)||"&grp1" as Response,   

   n1 label="Group&grp1 Sample Size (n1)",mean1 label="Group&grp1 mean",  

   sd1 label="Group&grp1 SD", 

   n2 label="Group&grp2 Sample Size (n2)", mean2 label="Group&grp2 Mean", 

   sd2 label="Group&grp2 SD",eff_size format=8.3 label="Effect Size (d)", 

   "("||trim(left(put(ci_low,8.3)))||","||trim(left(put(ci_up,8.3)))||")"  

    label="&conf % Conf. Interval"  

   from work.effectsize_table_IG; 

  quit; 

  title; 

 %end; 

 %if &groupvar=  %then %do; 

  title "Table 1: Effect Size (Cohen'd) and its Confidence Interval 

    for Paired or Repeated Measured(RM) Design"; 

  proc sql nonumber; 

  select  trim(Response) ||"&timeval2"||"-"||trim(Response)||"&timeval1" as 

Response,   

   n as n label="Sample Size (n)", mean1 label="Pre-test Mean", 

   sd1   label="Pre-test SD",mean2 label="Post-test Mean", 

sd2   label="Post-test SD", effect_size format=8.3 label="Effect Size 

(d)", 

   "("||trim(left(put(ci_low,8.3)))||","||trim(left(put(ci_up,8.3)))||")"  

    label=" &conf% Conf. Interval"  

   from work.effectsize_table_RM; 

  quit; 

  title; 

 %end; 

  

 %if &groupvar^= and &timevar^= %then %do; 

  title "Table 1: Effect Size (Cohen'd) and its Confidence Interval 

   for Independent Group Pretest Posttest Design(IGPP)"; 

  proc sql nonumber; 

  select * from work.effectsize_table_IGPP; 

  quit; 

  title; 

 %end; 

 ods  html close; 

%mend effect_size; 

 

**  Examples:Invoking macro to calculaate effect sizes **; 

*** Independent 2 group design ***; 

%effect_size( data=SESUG12.survey_data, 

   idvar=id, 

   groupvar=GROUP,  

   respvars=PCL BSI TRGI CESD, 

   cl=0.95) 

*** Repeated (sd calculated from indep. design) design ***; 

%effect_size(data=SESUG12.survey_data, 

   idvar=id, 

   design=1, /* 1 is 2-group independent,  2 is pre-post */ 

   groupvar=, /* or completely remove groupvar from list */ 

   timevar=VISIT,  

   timeval1=1, 

   timeval2=2,  

   respvars=PCL BSI TRGI CESD, 

   cl=0.95) 

** Repeated (sd calculated from change scores) design ***; 

%effect_size(data=SESUG12.survey_data, 

   idvar=id, 

   design=2, /* 1 is 2-group independent,  2 is pre-post */ 

   groupvar=, /* or completely remove groupvar from list */ 

   timevar=VISIT, 

   timeval1=1, 

   timeval2=2,  

   respvars=PCL BSI TRGI CESD, 

   cl=0.95) 

** Ind. Group Pre-Post design: (sd calculated from change scores) ***; 

%effect_size(data=SESUG12.survey_data, 

   idvar=id, 

   design=1, 

   groupvar=GROUP, 

   timevar=VISIT,  

   timeval1=1, 

   timeval2=2,  

   respvars=PCL BSI TRGI CESD, 

   cl=0.95) 

** Ind. Group Pre-Post design: : design=2 (sd calculated from change scores)*; 

%effect_size(data=SESUG12.survey_data, 

   idvar=id, 

   design=2, /* 1 is 2-group independent,  2 is pre-post */ 

   groupvar=GROUP, 

   timevar=VISIT,  

   timeval1=1, 

   timeval2=2,  

   respvars=PCL BSI TRGI CESD, 

   cl=0.95) 

 

 


